    COMPLETE TURBO BASIC DOCUMENTATIONPRIVATE 

             by Ron. Fetzer

FORWARD:
     This expanded documentation is based on the pioneering work of Dave and Laura Yearke of the Western New York Atari Users Group.    A quick summary and table of contents is also provided.  The information contained here is believed accurate, but the authors make no guarantees that you will not find errors.

REQUIREMENTS:
     The superb Turbo Basic by Frank

Ostrowski from Holland works only on

the Atari XL and Atari XE computers,

although another version has appeared

which purports to run on the 400/800

models.  This documentation only

applies to the XL/XE version.  Much of

it may be applicable to the other

version, but it has not been tested.

Turbo Basic will run with any DOS 2.0 or

2.5 workalike, but at this time, TURBO BASIC cannot be used with any version of SPARTA DOS.  As this is written in late 1992, a patched version that will run with SPARTA DOS has been reported to exist.  Hopefully, this version can be made to work well with SPARTA DOS.

FEATURES:
     TURBO BASIC is upwardly compatible

with ATARI BASIC.  In addition to the

regular Atari Basic commands there are

42 more commands and 22 more functions.

It has a number of compatible features (but is not directly compatible with)

BASIC XL and BASIC XE.  Useable RAM is increased to 34,021 bytes.  This is 1,747 more bytes than Atari Basic.  It runs 3 times faster than Atari Basic and has a BASIC COMPILER that increases the speed 3 to 5 times faster still (10-15X overall).

     The DOS commands are accessed in

the direct mode or the programmed mode. There is no Turbo Basic DOS in the conventional sense.  Typing "DOS" overwrites Turbo Basic and any program in memory will be lost.  For the quickest loading, place Turbo Basic on a RAMDISK. Loading will take just a few seconds.  A major advantages is in new commands that make structured programming possible. With modest planning, "GOTO" will seldom be needed.

     In addition, there are new graphics

commands not available before. Turbo

Basic is also insensitive to upper, lower

or inverse characters for most of its

commands. You can have 256 variable

names.  Since PROCEDURES can be named,

this increase in the allowable number

of variables is quite useful.

     This documentation tries to

provide programming examples for all

the Basic commands.  There is a table of

contents and a command summary to help

find information quickly.  In order to

distinguish the new features, all new

commands and functions will have a (T)

following their listing.

LOADING TURBO BASIC:
     Turbo Basic gets loaded without

a cartridge. 1. Turn on the disk drive.

2. Insert the Turbo Basic disk. 3. On the XE hold down the OPTION key and turn on the computer.

         ** DISK COMMANDS **
         -------------------

     The DOS functions can be accessed

in the direct or the programmed mode.

All regular commands are available except COPY.  There are several file copiers that run under Turbo Basic, so even the lack of a copy command poses no serious limitation.

     FORMAT:  BLOAD "D:name"  (T)

     EXAMPLE: BLOAD "D:TEST.BIN"

     This command is used to load (but not run) binary files. The file will automatically load without having to give a memory address. It is just like Option L with /N on DOS 2.0.  Note that a BINARY

LOAD can overwrite the programs already in memory.

     FORMAT:  BRUN "D:name"  (T)

     EXAMPLE: BRUN "D:COMPILER.COM"

     This is used to load and RUN a

machine language(Binary) file. The

file will run automatically without

having to give a memory address. It

is the same as Option L and Option M

on DOS 2.0.  Note the caution listed under BLOAD.

     FORMAT:  DELETE "D:name"  (T)

              (Wild cards are O.K.)

     EXAMPLE: DELETE "D:TEST.BAS"

              DELETE "D8:test.bas"

              delete "D2:*.*"

              Delete "D2:TEST.???"

              90 DELETE "D:*.*"

     This command erases from the

disk a file specified. The wild card

symbols (*.* and ???) are permitted.

The device name (D:) has to be in

capital letters. Inverse file names

are not permitted. In the programmed

mode it has to be the last item on

the line. This is the same as Option

D on DOS 2.0.

     FORMAT:  DIR  (T)

     (Note: Wild cards O.K.)

     EXAMPLE: DIR

              DIR "D:*.*"

              DIR "D2:*.BAS"

              DIR "D:???.*"

              DIR "D8:*.*"

              90 DIR "D8:*.*"

     This command gives you the disk

directory. Wild cards are permitted.  In the programmed mode it has to be the last item on the line. This is the same as Option A on DOS 2.0.  Note that as with Atari Basic, the device and filename will be accepted without the trailing quotation mark (").

     FORMAT:  RENAME "D:old name,new               name"  (T)

     EXAMPLE: RENAME "D:FILE.OLD,

              FILE.NEW"

              RENAME "D8:*.TXT,

              *.BAS"

     This command renames a file. The

first name is the OLD name, the second name is the NEW name. The comma between them is important. Wild cards are permitted. This is the same as Option E on DOS 2.0.

     FORMAT:  LOCK "D:name"  (T)

     EXAMPLE: LOCK "D:TEST.BAS"

              LOCK "D2:SCREEN.???"

              LOCK "D8:*.*"

              90 LOCK "D:*.*"

     This command locks a file so you

cannot change it or erase it until you unlock it. Wild cards are permitted. In the programmed mode it has to be the last item on the line.  This is the same as Option F on DOS 2.0.

     FORMAT:  UNLOCK "D:name"  (T)

     EXAMPLE: UNLOCK "D:TEST.BAS"

              UNLOCK "D2:SCREEN.???"

              UNLOCK "D8:*.*"

              10 UNLOCK "D:*.*"

     This command is the opposite of LOCK. It opens a file so it can be changed. Wild cards are permitted. In the programmed mode it has to be the last item on the line. This is the same as Option G on DOS 2.0

            ** MEMORY **
            ------------

     FORMAT:  DPOKE m,v  (T)

     (m=Memory Location, v=Value)

     EXAMPLE: DPOKE 88,32750

              90 DPOKE 88,32750

     This command puts 2 bytes of data into 2 consecutive memory locations. In the example above at POKE 88 and POKE 89 integer 32750 is stored.

     FORMAT:  DPEEK(m)  (T)

     EXAMPLE: ? DPEEK(88)

              90 ? DPEEK(88)

     This is the opposite of DPOKE.

It retrieves a 2 byte integer from 2

consecutive locations. In the example

above at PEEK(88) and PEEK(89)

     FORMAT:  MOVE m,m1,m2 (T)

     EXAMPLE: MOVE 53248,32768,1024

              MOVE $D000,$8000,$400

     This moves a whole block of

data. The 'm' is the old address, the

'm1' is the new address and the 'm2'

is the length of the data in bytes

     FORMAT:  -MOVE m,m1,m2 (T)

     EXAMPLE: -MOVE 53248,32768,1024

              -MOVE $D000,$8000,$400

     This is the same as above except

it copies with the last byte of the

block

     FORMAT:  BPUT #n,adr,len (T)

     EXAMPLE: BPUT #3,ADR(B$), LEN(B$)

     EXAMPLE #2:

     100 CLS

     110 GRAPHICS 5:ADDR=DPEEK(88)

     120 PRINT "    FILL THE SCREEN"

     130 FOR X = 1 TO (20*40)

     140 POKE ADDR+X,RAND(125)

     150 NEXT X

     160 ? "Now BPUT data onto

         RAMDISK D8:SCREEN"

     170 CLOSE:OPEN #1,8,0,"D8:

         SCREEN"

     180 BPUT #1,ADDR,20*40

     190 CLOSE

     200 ? "IT IS DONE!"

     210 FOR Y= 1 TO 6000:NEXT Y

     220 GRAPHICS 0

     This outputs a block of data to a device specified by the channel number(1- 7). The block starts at 'adr' and ends at 'len'. In the example above we wrote to the RAM DISK the screen data under the file name "SCREEN". It can be recovered

with BGET. See the next command.

     FORMAT:  BGET #n,adr,len (T)

     EXAMPLE: BGET #3,ADR(B$),

              LEN(B$)

     EXAMPLE #2:

     100 CLS

     110 GRAPHICS 5:ADDR=DPEEK(88)

     120 CLOSE:OPEN #1,4,0,

         "D8:SCREEN"

     130 ? "BGET data from RAM DISK

         D8:'SCREEN'"

     140 BGET #1,ADDR,20*40

     150 CLOSE

     160 FOR Y=1 TO 6000:NEXT Y

     170 GRAPHICS 0

     BGET recovers a block of data

saved with BPUT. The '#n' is the

channel number(1-7), 'adr' is the

memory address and 'len' is the

length of the data.

     FORMAT:  %PUT #n,a (T)

     EXAMPLE: %PUT #1,A

     EXAMPLE #2:

     10 CLS

     20 CLOSE:OPEN #1,8,0,"D8:TEST"

     30 FOR X = 1 TO 5

     40 INPUT "GIVE ME A NUMBER>",A

     50 %PUT #1;A

     60 NEXT X:CLOSE

     70 ?:? "I have saved the

        the numbers with %PUT to the

        RAM DISK under 'TEST'"

     80 ?:?:DIR "D8:*.*"

     This is the same as 'PRINT #1;A' except it is much faster and uses less memory. You must open a channel for %PUT. See the demonstration program on the front of this disk.  Use RUN "D:PUTGET.TUR". If you have a XL computer list the program and change all references from D8: to D1: and then run it. You can retrieve the numbers in EXAMPLE #2 by using the %GET program that follows. It is used with NUMBERS only.

     FORMAT:  %GET #n,a (T)

     EXAMPLE: %GET #1,A

     EXAMPLE #2:

     10 CLS

     20 CLOSE:OPEN #1,4,0,"D8:TEST"

     30 FOR Y = 1 TO 5

     40 %GET #1;A

     50 ? A

     60 NEXT Y:CLOSE

     70 ?:? "I have retrieved the

        numbers saved with %PUT by

        using %GET"

     80 LIST

     It retrieves the numbers saved

with %PUT. You must have a channel

open. It is the same as 'INPUT #1;A'

except it is much faster.

     **  GENERAL PROGRAMMING **
     ---------------------------

     FORMAT:  CLS (T)

     EXAMPLE: CLS

     EXAMPLE #2:

     10 CLS

     20 FOR X=1 TO 50

     30 ? " CLEAR SCREEN ";

     40 NEXT X

     50 GOTO 10

     This command clears the screen.

It is the same as pressing

CNTRL+CLEAR or ? CHR$(125)

     FORMAT:  CLS #n(channels 1-7) (T)

     EXAMPLE: CLS #3

              90 CLS #3

     EXAMPLE #2:

     10 CLS

     20 ?:?:?:?:?:?:? "THIS WILL NOT BE      ERASED - IT IS BEFORE LINE 30(OPEN).      TO STOP -->'BREAK'"

     30 CLOSE:OPEN #3,4,0,"K:"

     40 CLS #3

     50 ? "PRESS ANY KEY";

     60 GET #3,L

     70 ?:? "YOU PRESSED THE ";

        CHR$(L);" KEY":?

     80 CLS #3

     90 GOTO 40

     This command clears the screen

of any information that was there

because of the opened channel. This

is different from CLS which clears

the whole screen.

     FORMAT:  PAUSE n (T)

     EXAMPLE: PAUSE 50

              90 PAUSE 100

     EXAMPLE #2:

     10 CLS

     20 REPEAT:A=A+1

     30 ? A;", ";

     40 PAUSE 100

     50 UNTIL A=20

     EXAMPLE #3

     10 CLS

     20 X=10

     30 REPEAT:A=A+1

     40 ? A;", ";

     50 PAUSE 10*X

     60 UNTIL A=20

     This command puts a time delay

of about 1/60 of a second into your

program for each number. PAUSE 60 =

about 60/60 or 1 second delay. It is

the similar to 'FOR X = 1 TO 60:NEXT

X'. Computed values are allowed.

Negative values are not allowed.

     FORMAT:  RENUM n,i,j (T)

     EXAMPLE: RENUM 10,100,10

              90 RENUM 100,10,10

     EXAMPLE #2:

     10 CLS

     20 ? "RENUMBER PLEASE!"

     30 PAUSE 100

     40 LIST

     50 RENUM 10,100,20

     60 LIST

     EXAMPLE #3

     10 CLS

     20 X=3

     30 ? " HELLO ";

     40 GOTO X*10

     50 REM X*10 WILL NOT BE

        RENUMBERED

     60 RENUM 10,100,20

     70 LIST

     This command renumbers your

program. The 'n' is the program line

number that you want to start

renumbering with. The 'i' is the

first NEW number. The 'j' is the

increment. It will not renumber line

references which involve variables or

computed values. See example #3.

     FORMAT:  DEL n,i (T)

     EXAMPLE: DEL 60,100

              90 DEL 50,80

     EXAMPLE #2:

     10 CLS

     20 ? "THIS LINE STAYS"

     30 ? "30 GOES"

     40 ? "40 GOES"

     50 ? "50 GOES"

     60 DEL 30,50

     70 LIST

     This command deletes lines out

of your program. 'n' = the starting

number and 'i'= the ending number. It

can be used in the direct or

programmed mode.

     FORMAT:  DUMP (T)

     EXAMPLE: DUMP

              90 DUMP

     EXAMPLE #2:

     10 CLS

     20 DIM A$(35),B(5)

     30 A$="DUMP - VARIABLES AND

        THEIR VALUES"

     40 REPEAT:X=X+1

     50 INPUT "GIVE ME A NUMBER>",C

     60 B(X)=C:REM NUMBER ARRAY

     70 UNTIL X=5

     80 ? "   NUMBERS"

     90 FOR T = 1 TO 5

     100 ? B(T)

     110 NEXT T

     120 ?:? A$

     130 DUMP

     This command dumps the variables

used and their values. For numeric

arrays the DUMPed value will be shown

as: (B,6. The DUMPed value is the

DIMed value +1. String values are

shown with their length first and

then their DIMed value. PROCEDURE

names and labels and their line

values are also listed. The NEW

command clears the DUMPed values.

This command can be used in the

direct or the programmed mode.

     FORMAT:  DUMP name (T)

     EXAMPLE: DUMP "P:"

              DUMP "D8:VARIB"

              90 DUMP "P:"

     EXAMPLE #2:

     10 CLS

     20 DIM A$(40),B(5)

     30 A$="DUMPed - VARIABLES"

     40 FOR X = 1 TO 5

     50 INPUT "GIVE ME A NUMBER>",C

     60 B(X)=C:REM NUMBER ARRAY

     70 NEXT X

     80 FOR Y = 1 TO 5

     90 ? B(Y):LPRINT B(Y)

     100 NEXT Y

     110 LPRINT:LPRINT A$

     120 DUMP "P:"

     This command is the same as above except we DUMP to a device specified. "P:" = printer or "D8:VARIB" = Ram Disk. For example #2 turn on your printer. This command can be used in the direct or the

programmed mode.

     FORMAT:  TRACE (T)

              TRACE -

     EXAMPLE: 10 TRACE

              90 TRACE -

     EXAMPLE #2:

     10 TRACE

     20 CLS

     30 ?:? "   TRACING"

     40 REPEAT:A=A+1

     50 ? A;", ";

     60 UNTIL A=10

     70 TRACE -

     80 ?:?

     90 ? "   NO TRACING"

     100 REPEAT:B=B+1

     110 ? B;", ";

     120 UNTIL B=10

     This command tells you what line

is about to be executed. It is shown

in [ ] on the screen. The TRACE is

the way you turn it ON. The TRACE -

 is the way you turn it OFF. This is

a great aid in de-bugging a program.

     FORMAT:  INPUT "text";a,b... (T)

              INPUT "text >",a,b...

     EXAMPLE: 20 INPUT "Numbers";A

              50 INPUT "Names";B$

              70 INPUT "Names &"

                 Ages >>",A$,B

              90 INPUT "ADDRESS >"

                 ,ADR$

     EXAMPLE #2:

     10 CLS

     20 DIM A$(10)

     30 ? "THIS WILL PRINT ?":?

     40 FOR X = 1 TO 3

     50 INPUT "Names";A$

     60 NEXT X

     70 ?:?

     80 ? "No Symbol - Use your own"

        :?

     90 FOR Y = 1 TO 3

     100 INPUT "Numbers >>",B

     110 NEXT Y

     This command lets you use a

prompt with the INPUT for easier programming. If you use a semicolon after the prompt the usual '?' will be printed. If you use a comma after the prompt nothing will be printed and you have to include your own symbol with the prompt. See example #2.

     FORMAT:  GO TO

     EXAMPLE: 50 GO TO 20 (T)

              GO TO 100

     EXAMPLE #2:

     10 CLS

     20 ? " HELLO ";

     30 GO TO 20

     (This command is the same as "GOTO")

     FORMAT:  *L (DEFAULT) (T)

              *L -

     EXAMPLE: 30 *L

              90 *L -

     EXAMPLE #2:

     10 CLS:?

     20 *L

     30 ? "LINE # 50-70 INDENT ON"

     40 FOR X = 1 TO 5

     50 ? " ";

     60 ? " ";

     70 ? " ";

     80 NEXT X

     90 LIST

     100 PAUSE 300

     110 CLS:?

     120 *L -

     130 ? "LINE # 50-70 INDENT OFF"

     140 LIST

     When listing a program with loops this command either turns on the indent on the loops or turns them off. The *L turns it ON. This is the normal default condition. The *L - turns the indent OFF. This works not only on the screen but also on the printer.

     FORMAT:  *F (or *F+) (T)

              *F -

     EXAMPLE: 70 *F

              90 *F -

     EXAMPLE #2:

     10 CLS

     20 ? "ILLEGAL REVERSE LOOP":?

     30 FOR X = 2 TO 1

     40 ? X

     50 NEXT X

     60 PAUSE 200

     70 *F

     80 ?:? "ILLEGAL LOOP - NO

        EXECUTION - '*F'"

     90 FOR X = 2 TO 1

     100 ? X

     110 NEXT X

     120 PAUSE 200

     130 LIST

     Atari Basic has a bug. It allows execution of an illegal reverse loop. (FOR X=2 TO 1:? X:NEXT X). This command allows you to turn this feature off or on. *F or *F + turns it OFF. *F - turns the bug ON if you want it.

     FORMAT:  *B (or B+) (T)

              *B -

     EXAMPLE: 90 *B

              200 *B -

     EXAMPLE #2:

     10 CLS:?

     20 CLOSE:OPEN #1,8,0,"D8:TEST"

     30 FOR X=1 TO 3

     40 INPUT "GIVE ME A NUMBER";A

     50 %PUT #1;A

     60 NEXT X:CLOSE

     70 PAUSE 100:?:?

     80 ? " I AM TRAPPING 'BREAK'

        KEY"

     90 *B

     100 CLOSE:TRAP 180:

         OPEN #1,4,0,"D8:TEST"

     110 ?:?

     120 FOR Y = 1 TO 3

     130 PAUSE 100

     140 %GET #1;A

     150 ? A;" TRY BREAK KEY"

     160 NEXT Y:CLOSE

     170 END

     180 REM TRAP

     190 ?:? "BREAK KEY TRAPPED!"

     200 GOTO 90

     This command lets you trap the break key so your program cannot accidentally be stopped by hitting the BREAK key. This command works with the TRAP command. The *B turns it ON. The *B - turns it OFF.

     FORMAT:  -- (T)

     EXAMPLE: 10 --

              90 --

     EXAMPLE #2:

     10 --

     20 CLS

     30 ? " 30 DASHES "

     40 ? " HELLO "

     50 --

     60 LIST

     This command prints 30 dashes

across the screen or the page when

you LIST your program. It can be used

to make your program listing clearer.

     FORMAT:  DSOUND n,f,d,v (T)

              DSOUND

     EXAMPLE: 10 DSOUND 1,72,12,8

              90 DSOUND

     EXAMPLE #2:

     10 CLS

     20 ?:? "PRESS 'RESET' TO STOP"

     30 X = RAND(255)

     40 ? X;", ";

     50 DSOUND 1,X,12,8

     60 PAUSE 25

     70 DSOUND

     80 DSOUND 0,X,12,8

     90 GOTO 30

     This command pairs channels for

increased frequency range.

n=channels(0-3). f=frequency range(0-

255). d=distortion(0-15) 10 and 14

produce pure tones. v=volume(0-15). 0

is silent 15 is the loudest. DSOUND

turns all sound OFF.

         **  LINE LABELS  **
         -------------------

     FORMAT:  # name (T)

     EXAMPLE: 120 # END

              120 # END_OF_FILE

     EXAMPLE #2:

     10 CLS

     20 CLOSE:OPEN #1,8,0,"D8:TEST"

     30 REPEAT:X=X+%1

     40 INPUT "GIVE ME A NUMBER";A

     50 %PUT #1,A

     60 UNTIL X=%3

     70 CLOSE

     80 CLOSE:OPEN #1,4,0,"D8:TEST"

     90 TRAP # END:%GET #1,A

     100 ? A

     110 GOTO 90

     120 # END:IF ERR=136 THEN CLOSE

     This command gives the line number that has the '#' the VARIABLE LABEL that follows the '#'. For example: '120 # END will now become a VARIABLE LABEL and it will be in the variable name table listed as such (Use DUMP to see it). In example #2 see line #90 '90 TRAP # END...".  It will go to line 120. This is the same as if we wrote '90 TRAP 120...' except it is clearer the first way.  When there is an EOL error(136) channel #1 will be closed so no error is printed.  A better way would have been '120  # END_OF_FILE...' The underscore (_) is allowed in variables and labels and makes for clearer listings. We would have a better description of WHERE and WHAT the program does if we had used it.  If you use line labels they can be renumbered. %1 and %3 in line 30 and 60 are used because they save us 6 bytes each and stand for the number 1 and 3.  See the dicussion on constants.

     FORMAT:  GO# name (T)

     EXAMPLE: 60 GO# GET_KEY

              100 GO# ERROR

     EXAMPLE #2:

     10 CLS

     20 ?:? "PRESS ANY KEY"

     30 CLOSE:OPEN #1,4,0,"K:"

     40 # GET_KEY:GET #1,A

     50 ? CHR$(A);", ";

     60 GO# GET_KEY

     This is the same as GOTO except the GO# looks for a label. You cannot have a space between the GO and #.  For the (_) in line 40 see '# name' above.  Line 40 has the variable label of '# GET_KEY'. Line 60 looks for this label and it works just like a GOTO except the listing is clearer.

 * MODIFICATION OF ATARI COMMANDS *
 ----------------------------------

     FORMAT:  CLOSE (T)

     EXAMPLE: 90 CLOSE

              120 CLOSE:

              OPEN #1,4,0,"D8:TEST"

     This command closes ALL opened channels. It is a good practice to close a channel first before opening it. This command is the same as 'FOR X=1 TO 7:CLOSE X:NEXT X'

     FORMAT:  DIM a(n) (T)

     EXAMPLE: 10 DIM A(6)

              30 DIM B$(20)

     This command automatically assigns a value of 0 to all elements in a numeric array and null characters to all elements of a string. The LEN function is still

variable and initially 0.

     FORMAT:  GET name (T)

     EXAMPLE: 20 GET A

              90 GET K

     EXAMPLE #2:

     10 CLS:?:?

     20 GET A

     30 ? "The key is ";CHR$(A);

     40 ? " the ATASCII = ";A

     50 GOTO 20

     This command returns the ATASCII

value of the key pressed. It is the

same as 'OPEN #1,4,0,"K:":GET

#1;A:CLOSE #1'. The command waits for

a key press and assigns the ATASCII

value to the variable.

     FORMAT:  LIST n, (T)

     EXAMPLE: LIST 90,

     This command list the program

from the line specified to its end provided the comma is used.

     FORMAT:  ON a EXEC n1,n2,... (T)

     EXAMPLE: 30 ON X EXEC START,

              MAIN_PROG,ENDING

     This command is the same as

'ON...GOSUB' except it is used with

PROCEDURE. The command executes the

PROCEDURES named. Instead of 'ON X

GOSUB 200,300,400' we use variable

label names of the PROCEDURES.  See

PROCEDURE.

     FORMAT:  ON a GO# n1,n2,...

     EXAMPLE: 70 ON X GO# NORMAL,               UPSIDE_DOWN

     EXAMPLE #2:

     10 CLS:DIM A$(1)

     20 ?:?

     30 INPUT "Normal or Upside

        down";A$

     40 IF A$(1,1)="N" THEN X=1

     50 IF A$(1,1)<>"N" THEN X=2

     60 ?:?

     70 ON X GO# NORMAL,UPSIDE_DOWN

     80 # NORMAL

     90 POKE 755,2

     100 ? "THIS IS PRINTED NORMALLY"

     110 GOTO 20

     120 # UPSIDE_DOWN:CLS

     130 POKE 755,7:?:?

     140 ? "THIS IS PRINTED UPSIDE

         DOWN"

     150 PAUSE 200:POKE 755,2

     160 GOTO 20

     This command is the same as ON...GOTO except we use variable label names instead of line numbers.  In example #2 see line 70.

     FORMAT:  POP

     EXAMPLE: 100 POP

     EXAMPLE #2:

     10 CLS:?:?

     20 GOSUB 50

     30 ? "I am the next line"

     40 END

     50 REM SUBROUTINE

     60 POP

     70 GOTO 90

     80 RETURN

     90 ? "I POPPED the RETURN

        line 30"

     100 LIST

     The POP command will work with all 4 loops, the DO-LOOP, WHILE-WEND, FOR-NEXT and GOSUB. When TURBO BASIC sees a DO, WHILE, FOR or GOSUB it saves its RETURN address on the runtime stack so it knows to what line to return to. If you want BASIC to 'FORGET' this return line use POP.  Too many POPs can lead to errors and make for poor programming.

     FORMAT:  PUT n (T)

     EXAMPLE: 60 PUT 90

              20 PUT 89,69,83

     EXAMPLE #2:

     10 REM SIMPLE WORDPROCESSOR

     20 CLS:?

     30 OPEN #1,4,0,"K:"

     40 GET #1;A:R=R+1

     50 IF R=37 THEN ? CHR$(155):R=0

     60 PUT A

     70 GOTO 40

     This command is the same as 'PRINT CHR$(A)';. It takes the ATASCII number and converts it into a character.

     FORMAT:  RESTORE #name (T)

     EXAMPLE: 120 RESTORE #AGAIN

              160 RESTORE #NUM_AGAIN

     EXAMPLE #2:

     10 CLS:?

     20 DATA 1,2,3,4,5

     30 # AGAIN

     40 DATA 6,7,8,9,0

     50 READ A

     60 IF A=0 THEN 100

     70 ? A;", ";

     80 PAUSE 10

     90 GOTO 50

     100 ?:? "NOW DOING LN 40 DATA

        AGAIN"

     110 PAUSE

     120 RESTORE #AGAIN

     130 READ A

     140 IF A=0 THEN 170

     150 ? A;", ";

     160 GOTO 130

     170 ?:LIST

     This command moves the data pointer back to the variable label in the DATA chain. In example #2 the RESTORE printed only the numbers in DATA ln. 40 because it came after '30 #AGAIN'. It did not start with ln.  #20. It is the same as RESTORE 40 except we use a variable label for clearer programming. The 0 was the

flag. When the READ saw a 0 then the IF- THEN came in and directed it to some other line.

     FORMAT:  RND

     EXAMPLE: 20 RND (T)

              40 X=RND

              90 X=INT(RND*10)

              100 X=INT(RND*10)+1

              200 X=INT(RND*100)

     EXAMPLE #2:

     10 X=RND

     20 ? X;", ";

     30 GOTO 10

     EXAMPLE #3:

     10 CLS:?

     20 X=INT(RND*10)

     30 PAUSE 25:? X;", ";

     40 GOTO 20

     In this command the dummy variable (0) has been eliminated. You can still use the ATARi way RND(0) if you want to. In the example ln. 90 you get random numbers from 0-9. In ln. 100 you get random numbers from 1-10. In line 200 you get random numbers from 0-99

     FORMAT:  RAND(n) (T)

     EXAMPLE: 10 X = RAND(10)

              90 X = RAND(100)

     EXAMPLE #2:

     10 CLS

     20 X = RAND(100)

     30 ? X;", ";

     40 PAUSE 25

     50 GOTO 20

     In this command the random numbers are printed as integers. They start with "0" and end with the number- 1 in the parentheses.

     FORMAT:  SOUND

     EXAMPLE: 90 SOUND

     EXAMPLE #2

     10 R=R+1

     20 Y = RAND(255)

     30 SOUND 0,Y,10,8

     40 PAUSE 100

     50 SOUND

     60 X = RAND(255)

     70 SOUND 1,X,12,8

     80 PAUSE 50

     90 SOUND

     100 IF R=8 THEN SOUND:

         R=0:GOTO 120

     110 GOTO 10

     120 END

     This command turns all sounds off. 

     FORMAT:  TRAP #name

     EXAMPLE: 90 TRAP #END

              95 TRAP #END_OF_FILE

     This command is the same as '90 TRAP 150' except it uses variable label names. For an application see example #2 on the command  "#name".

